ORIGINAL (論　文)

Numerical Analysis of Orthotropic Plates (I)*

Minoru MASUDA**, Hikaru SASAKI** and Takamaro MAKU**

増田　稔**・佐々木　光**・満久崇麿**：直交異方性板の数値解析（I）*

Introduction

The solutions of many problems of orthotropic plates are literally impossible when attempted by applying the differential equations of the theory of elasticity. To solve such problems mathematical and physical approximate methods must be used. The former reminds us of the finite difference method[1),2)] and the latter, of the finite element method[3)]. Recently, the electronic computers make these numerical methods possible. In the present paper, applications of the finite difference method to the bending problems of orthotropic layered plates are attempted.

Theory of Layered Orthotropic Plates

1. The Fundamental Equation of Orthotropic Layered Plates

The differential equation for the deflection of layered orthotropic plates is derived by the following procedures:

The equations of equilibrium are

$$\frac{\partial Q_x}{\partial x}+\frac{\partial Q_y}{\partial y}+p=0, \tag{1—1}$$

$$Q_x=\frac{\partial M_x}{\partial x}+\frac{\partial M_{yx}}{\partial y}, \tag{1—2}$$

$$Q_y=\frac{\partial M_y}{\partial y}+\frac{\partial M_{xy}}{\partial x}, \tag{1—3}$$

and

$$M_{xy}=M_{yx}. \tag{1—4}$$

For the small deflections the strain-displacement relations are

$$u=-z\frac{\partial w}{\partial x}, \tag{2—1}$$

$$v=-z\frac{\partial w}{\partial y}, \tag{2—2}$$

* Presented at the 18th Annual Meeting of the Japan Wood Research Society, Kyoto, April, 1968 and reported briefly in J. Japan Wood Res. Soc., **14**, 441 (1968).

** Division of Composite Wood (木質材料研究部門).

Table 1. List of symbols.

Symbol	Definition
E	modulus of elasticity
G	modulus of rigidity
μ	Poisson's ratio
I	moment of inertia
M	bending moment
Q	shearing force
a, b	plate dimensions in x and y directions, respectively
p	applied pressure
t	thickness of the plate
u, v	components of displacements in x and y directions, respectively
w	deflection of plates
x, y	rectangular coordinates
z	distance from the neutral axis of the plate
ε	strain
γ	shearing strain
σ	normal stress
τ	shearing stress
θ	angle between x axis and X axis
$_{x, y}$ (subscript)	in x and y directions, respectively
$_{X, Y}$ (,,)	in X and Y directions, respectively (X, Y : principal elastic axes)
$_{v}$ (,,)	in veneer
$_{b}$ (,,)	in bending of the layered plate
$_{n}$ (,,)	in n-th lamina



$$\varepsilon_x = \frac{\partial u}{\partial x} = -z\frac{\partial^2 w}{\partial x^2}, \tag{2—3}$$

$$\varepsilon_y = \frac{\partial v}{\partial y} = -z\frac{\partial^2 w}{\partial y^2}, \tag{2—4}$$

$$\gamma_{xy} = \frac{\partial u}{\partial y} + \frac{\partial v}{\partial x} = -2z\frac{\partial^2 w}{\partial x \partial y}. \tag{2—5}$$

Assuming a state of generalized plane stress in the n-th lamina of the plate, the stress-strain relations are

$$\sigma_{xn}=\alpha_{11n}\varepsilon_x+\alpha_{12n}\varepsilon_y+\alpha_{16n}\gamma_{xy}\ , \tag{3—1}$$

$$\sigma_{yn}=\sigma_{21n}\varepsilon_x+\alpha_{22n}\varepsilon_y+\alpha_{26n}\gamma_{xy}\ , \tag{3—2}$$

$$\tau_{xyn}=\alpha_{61n}\varepsilon_x+\alpha_{62n}\varepsilon_y+\alpha_{66n}\gamma_{xy}\ , \tag{3—3}$$

and

$$\alpha_{12n}=\alpha_{21n},\ \alpha_{16n}=\alpha_{61n},\ \alpha_{26n}=\alpha_{62n}\ . \tag{3—4}$$

The moment resultants are formulated by the integration of (3-1)～(3-3) over each lamina and summing the resulting expressions over all laminas. Thus:

$$M_x=\sum_n\int_{t_{n-1}}^{t_n}\sigma_{xn}zdz$$

$$=\sum_n I_n(\alpha_{11n}\varepsilon_x+\alpha_{12n}\varepsilon_y+\alpha_{16n}\gamma_{xy})\ , \tag{4—1}$$

$$M_y=\sum_n I_n(\alpha_{12n}\varepsilon_x+\alpha_{22n}\varepsilon_y+\alpha_{26n}\gamma_{xy})\ , \tag{4—2}$$

$$M_{xy}=\sum_n I_n(\alpha_{16n}\varepsilon_x+\alpha_{26n}\varepsilon_y+\alpha_{66n}\gamma_{xy})\ . \tag{4—3}$$

Substituting (2-3)～(2-5) in (4-1)～(4-3) the following expressions are obtained:

$$M_x=-I\left(\beta_{11}\frac{\partial^2 w}{\partial x^2}+\beta_{12}\frac{\partial^2 w}{\partial y^2}+2\beta_{16}\frac{\partial^2 w}{\partial x\partial y}\right), \tag{5—1}$$

$$M_y=-I\left(\beta_{12}\frac{\partial^2 w}{\partial x^2}+\beta_{22}\frac{\partial^2 w}{\partial y^2}+2\beta_{26}\frac{\partial_2 w}{\partial x\partial y}\right), \tag{5—2}$$

$$M_{xy}=-I\left(\beta_{16}\frac{\partial^2 w}{\partial x^2}+\beta_{26}\frac{\partial^2 w}{\partial y^2}+2\beta_{66}\frac{\partial^2 w}{\partial x\partial y}\right), \tag{5—3}$$

where,

$$I=\sum_n I_n=\frac{t^3}{12}, \tag{5—4}$$

$$\beta_{ij}=I^{-1}\sum_n\alpha_{ijn}I_n\ . \tag{5—5}$$

From the equations of equilibrium (1-1)～(1-4), the following differential equation is obtained:

$$\frac{\partial^2 M_x}{\partial x^2}+2\frac{\partial^2 M_{xy}}{\partial x\partial y}+\frac{\partial^2 M_y}{\partial y^2}+p=0\ . \tag{6—1}$$

The substitution of (5-1)～(5-3) in (6-1) yields the differential equation for the deflection of layered orthotropic plates:

$$\beta_{11}\frac{\partial^4 w}{\partial x^4}+4\beta_{16}\frac{\partial^4 w}{\partial x^3\partial y}+(2\beta_{12}+4\beta_{66})\frac{\partial^4 w}{\partial x^2\partial y^2}$$

$$+4\beta_{26}\frac{\partial^4 w}{\partial x\partial y^3}+\beta_{22}\frac{\partial^4 w}{\partial y^4}=\frac{p}{I}\ . \tag{6—2}$$

2. *Relation of Elastic Properties between Veneer and Plywood*

Assuming plywood as a layered orthotropic material, the differential equation for the deflection is expressed in eq. (6-2), whose coefficients consist of the elastic constants of plywood. The relation between the elastic constants of veneer and plywood is given below :

$$E_{Xb}I=\sum_n E_{Xvn}I_n \ , \qquad (7\text{—}1)$$

$$E_{Yb}I=\sum_n E_{Yvn}I_n \ , \qquad (7\text{—}2)$$

$$G_{XYb}I=\sum_n G_{XYvn}I_n \ , \qquad (7\text{—}3)$$

$$\mu_{XYb}E_{Yb}I=\mu_{YXb}E_{Xb}I=\sum_n \mu_{XYvn}E_{Yvn}I_n=\sum_n \mu_{YXvn}E_{Xvn}I_n \ . \qquad (7\text{—}4)$$

When the principal axes of veneer coincide with the axes of coordinates, the coefficients α_{ijn}, are given in simply forms, and they are denoted with C_{ijvn} which are expressed as follows:

$$C_{11vn}=E_{Xvn}/\lambda_n \ , \quad C_{22vn}=E_{Yvn}/\lambda_n \ ,$$

$$C_{12vn}=\mu_{XYvn}E_{Yvn}/\lambda_n=\mu_{YXvn}E_{Xvn}/\lambda_n \ ,$$

$$C_{66vn}=G_{XYvn}, \quad \lambda_n=1-\mu_{XYvn}\mu_{YXvn} \ . \qquad (8\text{-}1)\sim(8\text{-}5)$$

α_{ijn} in the arbitrary direction expressed with C_{ijvn} are given by

$$\alpha_{11n}=C_{11vn}\cos^4\theta+C_{22vn}\sin^4\theta+2(C_{12vn}+2C_{66vn})\cos^2\theta\sin^2\theta \ , \qquad (9\text{—}1)$$

$$\alpha_{12n}=(C_{11vn}+C_{22vn}-4C_{66vn})\cos^2\theta\sin^2\theta+C_{12vn}(\cos^4\theta+\sin^4\theta) \ , \qquad (9\text{—}2)$$

$$\alpha_{16n}=(C_{11vn}\cos^2\theta-C_{22vn}\sin^2\theta)\cos\theta\sin\theta-(C_{12vn}+2C_{66vn})(\cos^2\theta-\sin^2\theta)\cos\theta\sin\theta \ , \qquad (9\text{—}3)$$

$$\alpha_{22n}=C_{11vn}\sin^4\theta+C_{22vn}\cos^4\theta+2(C_{12vn}+2C_{66vn})\cos^2\theta\sin^2\theta \ , \qquad (9\text{—}4)$$

$$\alpha_{26n}=(C_{11vn}\sin^2\theta-C_{22vn}\cos^2\theta)\cos\theta\sin\theta+(C_{12vn}+2C_{66vn})(\cos^2\theta-\sin^2\theta)\cos\theta\sin\theta \ , \qquad (9\text{—}5)$$

$$\alpha_{66n}=(C_{11vn}+C_{22vn}-2C_{12vn})\cos^2\theta\sin^2\theta+C_{66vn}(\cos^2\theta-\sin^2\theta)^2 \ . \qquad (9\text{—}6)$$

Substituting (9-1)$\sim$(9-6) in (5-5), the following is obtained for the coefficients of the fundamental equation of plywood :

$$\beta_{11}=I^{-1}\sum_n C_{11vn}I_n\cos^4\theta+I^{-1}\sum_n C_{22vn}I_n\sin^4\theta+2I^{-1}(\sum_n C_{12vn}I_n+2\sum_n C_{66vn}I_n)\cos^2\theta\sin^2\theta$$

$$=I^{-1}\sum_n\frac{E_{Xvn}I_n}{\lambda_n}\cos^4\theta+I^{-1}\sum_n\frac{E_{Yvn}I_n}{\lambda_n}\sin^4\theta+2I^{-1}\left(\sum_n\frac{\mu_{XYvn}E_{Yvn}I_n}{\lambda_n}+2\sum_n G_{XYvn}I_n\right)\cos^2\theta\sin^2\theta$$

$$=C_{11b}\cos^4\theta+C_{22b}\sin^4\theta+2(C_{12b}+2C_{66b})\cos^2\theta\sin^2\theta \ , \qquad (10\text{—}1)$$

where,
$$C_{11b}=I^{-1}\sum_n E_{Xvn}I_n/\lambda_n\,, \tag{10—2}$$

$$C_{22b}=I^{-1}\sum_n E_{Yvn}I_n/\lambda_n\,, \tag{10—3}$$

$$C_{12b}=I^{-1}\sum_n \mu_{XYvn}E_{Yvn}I_n/\lambda_n=\sum_n \mu_{YXvn}E_{Xvn}I_n/\lambda_n\,, \tag{10—4}$$

$$C_{66b}=I^{-1}\sum_n G_{XYvn}I_n\,. \tag{10—5}$$

β_{12}, β_{16}, β_{22}, β_{26} and β_{66} are also expressed with C_{11b}, C_{22b}, C_{12b} and C_{66b} in the same manner.

When plywood consists of veneers of a single species, the substitution of (7-1)~(7-4) in (10-2)~(10-5), yields the following relation:

$$C_{11b}=E_{Xb}/\lambda,\ C_{22b}=E_{Yb}/\lambda,\ C_{12b}=\mu_{XYb}E_{Yb}/\lambda=\mu_{YXb}E_{Xb}/\lambda,\ C_{66b}=G_{XYb}, \tag{11—1}$$

where,
$$\lambda=\lambda_n=1-\mu_{XYv}\mu_{YXv}\,.$$

Then, the coefficients of the differential equation of the plywood consisting of a single species are expressed with the nominal bending elastic constants (E_{Xb}, E_{Yb}, G_{XYb} and μ_{XYb}), but the Poisson's ratios of the veneers must be used for λ.

Application of the Finite Difference Method

1. Derivation of Finite Difference Equations

As the rigorous solution of the differential equation (6-2) cannot be obtained when β_{16} or β_{26} is not zero, the authors applied the finite difference method to solve it approximately. The replacement procedure of the differential equation by the corresponding finite difference equations is shown below. The first and second derivatives of w at a nodal point K (see Fig. 1) are expressed as follows:

$$\left.\frac{\partial w}{\partial x}\right|_K\approx\frac{1}{2h_x}(w_{i+1,j}-w_{i-1,j})\,, \tag{12—1}$$

$$\left.\frac{\partial w}{\partial y}\right|_K\approx\frac{1}{2h_y}(w_{i,j+1}-w_{i,j-1})\,, \tag{12—2}$$

$$\left.\frac{\partial^2 w}{\partial x^2}\right|_K\approx\frac{1}{h^2_x}(w_{i+1,j}-2w_{ij}+w_{i-1,j})\,, \tag{12—3}$$

$$\left.\frac{\partial^2 w}{\partial y^2}\right|_K\approx\frac{1}{h^2_y}(w_{i,j+1}-2w_{ij}+w_{i,j-1})\,, \tag{12—4}$$

$$\left.\frac{\partial^2 w}{\partial x\partial y}\right|_K\approx\frac{1}{h_xh_y}(w_{i+1,j+1}-w_{i+1,j-1}-w_{i-1,j+1}+w_{i-1,j-1})\,. \tag{12—5}$$

And the third and fourth derivatives are also expressed in the same way. For example,

$$\left.\frac{\partial^3 w}{\partial x^3}\right|_K \approx \frac{1}{2h_x^{\,3}}(w_{i+2,j}-2w_{i+1,j}+2w_{i-1,j}-w_{i-2,j})\,, \qquad (12\text{—}6)$$

$$\left.\frac{\partial^3 w}{\partial x^2 \partial y}\right|_K \approx \frac{1}{2h_x^{\,2}h_y}(w_{i+1,j+1}-2w_{i,j+1}+w_{i-1,j+1}-w_{i+1,j-1}+2w_{i,j-1}-w_{i-1,j-1})\,, \qquad (12\text{—}7)$$

$$\left.\frac{\partial^4 w}{\partial x^4}\right|_K \approx \frac{1}{h_x^{\,4}}(w_{i+2,j}-4w_{i+1,j}+6w_{ij}-4w_{i-1,j}+w_{i-2,j})\,, \qquad (12\text{—}8)$$

$$\left.\frac{\partial^4 w}{\partial x^3 \partial y}\right|_K \approx \frac{1}{4h_x^{\,3}h_y}(w_{i+2,j+1}-2w_{i+1,j+1}+2w_{i-1,j+1}-w_{i-2,j+1}-w_{i+2,j-1}+2w_{i+1,j-1}-2w_{i-1,j-1}+w_{i-2,j-1})\,, \qquad (12\text{—}9)$$

$$\left.\frac{\partial^4 w}{\partial x^2 \partial y^2}\right|_K \approx \frac{1}{h_x^{\,2}h_y^{\,2}}(w_{i+1,j+1}-2w_{i,j+1}+w_{i-1,j+1}-2w_{i+1,j}+4w_{ij}-2w_{i-1,j}+w_{i+1,j-1}-2w_{i,j-1}+w_{i-1,j})\,. \qquad (12\text{—}10)$$

The Fundamental Equation

Substituting the relations (12-8)~(12-10) in eq. (6-2) at a nodal point K, the fundamental equation is replaced by the corresponding finite difference equation,

$$\begin{aligned}&Ew_{i+1,j+2}+Bw_{i,j+2}-Ew_{i-1,j+2}\\ &+Dw_{i+2,j+1}+Hw_{i+1,j+1}+Sw_{i,j+1}+Gw_{i-1,j+1}-Dw_{i-2,j+1}\\ &+w_{i+2,j}+Rw_{i+1,j}+Fw_{ij}+Rw_{i-1,j}+w_{i-2,j}\\ &-Dw_{i+2,j-1}+Gw_{i+1,j-1}+Sw_{i,j-1}+Hw_{i-1,j-1}+Dw_{i-2,j-1}\\ &-Ew_{i+1,j-2}+Bw_{i,j-2}+Ew_{i-1,j-2}\approx Cp_K\end{aligned} \qquad (13\text{—}1)$$

where, $A=(2\beta_{12}+4\beta_{66})/(\beta_{11}\gamma^2)$, $B=\beta_{22}/(\beta_{11}\gamma^4)$,

$C=h_x^{\,4}/(I\beta_{11})$, $D=\beta_{16}/(\beta_{11}\gamma^3)$,



Fig. 1. Grid and the nodal points.



Fig. 2. Equilibrium equation expressed by the finite difference method.

$E=\beta_{26}/(\beta_{11}\gamma^3)$, $\quad F=4A+6B+6$,

$G=A+2D+2E$, $\quad H=A-2D-2E$,

$R=-2A-4$, $\quad S=-2A-4B$,

$\gamma=h_y/h_x$.

And this equation is expressed in the form shown in Fig. 2.

Boundary Conditions

Three typical edge conditions of the plates are,

i) simply supported edges——$(w=0,\ M_x=0)$ or $(w=0,\ M_y=0)$,

ii) clamped edges——$\left(w=0,\ \frac{\partial w}{\partial x}=0\right)$ or $\left(w=0,\ \frac{\partial w}{\partial y}=0\right)$,

iii) free edges——$(M_x=0,\ V_x=0)$ or $(M_y=0,\ V_y=0)$,

where,

$$V_x=Q_x+\frac{\partial M_{xy}}{\partial y}$$

$$=-I\left(\beta_{11}\frac{\partial^3 w}{\partial x^3}+4\beta_{16}\frac{\partial^3 w}{\partial x^2\partial y}+(\beta_{12}+4\beta_{66})\frac{\partial^3 w}{\partial x\partial y^2}+2\beta_{62}\frac{\partial^3 w}{\partial y^3}\right), \quad (14\text{—}1)$$

$$V_y=Q_y+\frac{\partial M_{yx}}{\partial x}$$

$$=-I\left(2\beta_{16}\frac{\partial^3 w}{\partial x^3}+(\beta_{12}+4\beta_{66})\frac{\partial^3 w}{\partial x^2\partial y}+4\beta_{26}\frac{\partial^3 w}{\partial x\partial y^2}+\beta_{22}\frac{\partial^3 w}{\partial y^3}\right), \quad (14\text{—}2)$$

(substituting (1-2)～(1-4) and (5-3)).

These edge conditions can also be replaced by the finite difference equations in the same manner as in the fundamental equation, and some of them are shown, as examples, in Figs. 3 and 4.

2. *Computation Procedure*

Now, we would like to explain the computation procedure of deflection of the orthotro-



Fig. 3. M_x expressed by the finite difference method.
$R_{11}=4\beta_{11}+4\beta_{12}\gamma^{-2}$, $R_{12}=-2\beta_{12}\gamma^{-2}$,
$R_{13}=-2\beta_{11}$, $R_{14}=\beta_{16}\gamma^{-1}$, $R_{1\bar{4}}=-R_{14}$.



Fig. 4. V_y expressed by the finite difference method.
$Q_{21}=\beta_{22}\gamma^{-3}$, $Q_{22}=2\beta_{16}$, $Q_{23}=(\beta_{12}+4\beta_{66})\gamma^{-1}+4\beta_{26}\gamma^{-2}$,
$Q_{24}=-(\beta_{12}+4\beta_{66})\gamma^{-1}+4\beta_{26}\gamma^{-2}$, $Q_{25}=4\beta_{16}+8\beta_{26}\gamma^{-2}$,
$Q_{26}=2(\beta_{12}+4\beta_{66})\gamma^{-1}+2\beta_{22}\gamma^{-3}$, $Q_{2\bar{i}}=-Q_{2i}$.



Fig. 5. Application of the finite difference method to the deflection problem of the orthotropic plate with a free edge.

pic rectangular plates giving an example of a plate with a free edge and three simply supported edges (see Fig. 5). In the figure, the finite difference equation of the fundamental equation holds at each nodal point of symbols (○) and (⊗), and that of the boundary condition holds at each point (⊗), (△) and (×).

In order to solve the simultaneous linear equations made of the above finite difference equations, the number of the equations must be equal to that of the unknowns.

In this case, for example, ten independent equations, that is, the deflections of eight imaginary nodal points on the extended lines of simply supported edges are equal to zero, and $w_1=-w_2$ and $w_3=-w_4$ must be added.

Thus, deflections at nodal points are obtained by solving the simultaneous linear equations. To improve the accuracy of the approximation, a fine grid is required. The finer the grid, the more complicate the matrix of the simultaneous linear equations becomes. The authors used the computer not only to solve the equations but also to make the matrix of the equations. In the computer programs the mesh sizes can be easily varied only by changing the input data for the purpose of extrapolation of the solution.

An example of the flow chart and the program in FORTRAN IV is shown in Appendix.

Experimental Procedure

As an example, measurement of the deflection of plywood under hydrostatic pressure was done.

1. Specimen

Five-ply of 1.58 : 2.34 : 1.58 : 2.34 : 1.58 (mm) red lauan veneer construction, 90 cm by 180 cm plywoods (9.4 mm thick) were made in the Hokkaido Forest Products Institute. From these the specimens were cut to 63 cm by 63 cm square with the grain of the face veneer oriented at 0° (or 90°), 15° (or 75°), 30° (or 60°) and 45° to the edges and conditioned for at least a year in the test room. The average of their moisture content and specific gravity were 11.5 % and 0.65, respectively. After deflection tests were done, 5 cm by 50 cm strips with the face grain oriented at 0°, 45° and 90° to the long side, were cut out of each square specimen and the elastic modulus in bending was measured with these strips.

2. Test Method

Apparatus for the deflection test is shown in Fig. 6 and Photo 1. Tests of simply supported plates and clamped plates can be done at the left and right part of the apparatus shown in the figure, respectively. For the ideal simply supported edge condition small wooden wedges were put between the specimen and the steel angle frame to set the specimen, and after pouring water into the vinyl bag, they were taken off. However, the ideal clamped edge condition seems to be difficult to perform in the experiment. The deflections were measured with dial gauges of an accuracy of 0.01 mm.



Fig. 6. Cross-sectional view of apparatus.
① Steel plate with 10 mm$^{\phi}$ rod, which is removed for a free edge
② Dial gauge
③ Specimen simply supported
④ Steel angle frame with 10 mm$^{\phi}$ rod
⑤ Specimen clamped
⑥ Vinyl bag
⑦ Steel angle frame for clamped edges



Photo. 1. Test apparatus.

Results and Discussions

1. Accuracy of the Solutions by the Finite Difference Method

As the finite difference method is a mathematical approximation method, the accuracy of the solution is one of the most important problems.

When β_{16} and β_{26} in the fundamental equation (6-2) are equal to zero, that is to say when the coordinate axes coincide with the principal elastic axes, the differential equation can be solved rigorously by using the Fourier series.

For example, the center deflection of the orthotropic rectangular plate with four simply supported edges under a concentrated load at the center is expressed as follows :[4),5)]

$$w_{center}=\frac{a^2mP}{\pi^3\sqrt{1-k^2}\,D_X}\sum_{n}^{1,3,\cdots}\frac{1}{n^3}\cdot\frac{h_2\sinh 2\alpha_1-h_1\sin 2\alpha_2}{\cosh 2\alpha_1+\cos 2\alpha_2}, \tag{15—1}$$

where,

$$m=\sqrt[4]{\frac{E_X}{E_Y}},\quad h_1=\sqrt{\frac{1+k}{2}},\quad h_2=\sqrt{\frac{1-k}{2}},$$

$$\alpha_1=\frac{bm\varphi}{2}h_1,\quad \alpha_2=\frac{bm\varphi}{2}h_2,\quad \varphi=\frac{n\pi}{a},$$

$$k=\frac{K}{\sqrt{D_XD_Y}},\quad D_X=\frac{E_{Xb}t^3}{12(1-\mu_{XYv}\mu_{YXv})},$$

$$D_Y=\frac{E_{Yb}t^3}{12(1-\mu_{XYv}\mu_{XYv})},\quad K=\mu_{YXb}D_X+D_{XY}$$

$$D_{XY}=\frac{G_{XYb}t^3}{6},\quad P:\text{ concentrated load}$$

The center deflection of a plate having the elastic constants of Table 3, was com-

Table 2. Accuracy of the solutions computed by the finite difference method. (Square orthotropic plate* with simply supported edges under a concentrated load)

	w_i	: w_{center}/P ($\times 10^{-3}$ cm/kg)	Ratio of w_i/w_0
Solution computed by eq. (15−1), $n=57$	w_0	1.5143	1.000
$h_x=a/4$, $h_y=a/4$	w_1	1.7814	1.176
$h_x=a/6$, $h_y=a/6$	w_2	1.6598	1.096
$h_x=a/8$, $h_y=a/8$	w_3	1.6066	1.061
$h_x=a/10$, $h_y=a/10$	w_4	1.5785	1.042
$h_x=a/12$, $h_y=a/12$	w_5	1.5618	1.031
$h_x=a/16$, $h_y=a/16$	w_6	1.5436	1.020

Extrapolation

Values used in the extrapolation	Extrapolated value : w_e ($\times 10^{-3}$ cm/kg)	Ratio of w_e/w_0
w_1 and w_2	1.5625	1.032
w_1 and w_3	1.5483	1.022
w_2 and w_3	1.5382	1.016
w_2 and w_4	1.5328	1.012
w_3 and w_4	1.5285	1.009
w_5 and w_6	1.5202	1.004

* the list of data of the elastic constants is shown in Table 3.

puted by the finite difference method for various mesh sizes, and compared with that computed from eq. (15-1). These are shown in Table 2. It is obvious that the difference between them diminishes with the mesh size and, for instance, becomes smaller than 5% at the mesh size of one-tenth of the edge length.

According to the mathematical theory of the finite difference method,[1),2)] the difference between the rigorous solution w_r and the approximate one w_{hi} can be expressed as follows:

$$w_r - w_{hi} = h_i^2\varphi_1(x, y) + h_i^4\varphi_2(x, y) + \cdots, \quad (16-1)$$

Table 3. List of data.

a	:	60	(cm)
b	:	60	(cm)
t	:	0.937	(cm)
E_{Xb}	:	92.1	($\times 10^3$ kg/cm^2)
E_{Yb}	:	39.1	($\times 10^3$ kg/cm^2)
G_{XYb}	:	5.0	($\times 10^3$ kg/cm^2)
μ_{XYb}	:	0.069	
E_{Xv}	:	127.1	($\times 10^3$ kg/cm^2)
E_{Yv}	:	4.1	($\times 10^3$ kg/cm^2)
G_{XYv}	:	5.0	($\times 10^3$ kg/cm^2)
μ_{XYv}	:	0.561	



Fig. 7. Influence of mesh sizes on the deflection along the center vertical line of the orthotropic square plate with the elastic constants of Table 3 and the maximum principal axis parallel to the horizontal edges under hydrostatic pressure (top edge free and others simply supported).



Fig. 8. Influence of mesh sizes on the deflection along the center vertical line of the orthotropic square plate with the elastic constants of Table 3 and the maximum principal axis oriented at 45° to the horizontal edges under hydrostatic pressure (top edge free and others simply supported).



Fig. 9. Influence of mesh sizes on the deflection along the center vertical line of the orthotropic square plate with the elastic constants of Table 3 and the maximum principal axis oriented at 0° to the horizontal edges under hydrostatic pressure (top edge free and others clamped).

where, $h_i = h_{xi} = h_{yi}$.

Substituting the approximate solutions (w_{h1} and w_{h2}) of the different mesh sizes (h_1 and h_2) and neglecting the terms of the higher order of the right hand side, we can extrapolate the solution by the following equation :

$$w_e = \frac{h_1^2 w_{h2} - h_2^2 w_{h1}}{h_1^2 - h_2^2}. \qquad (16\text{—}2)$$

Using this equation of the extrapolation, accuracy is easily improved as shown in Table 2.

Figs. 7, 8 and 9 are the examples of computation of deflections of plates with a free edge by the finite difference method. In these cases, the rigorous solution is very difficult to obtain, but the accuracy can be estimated by the extrapolation from the computed results of different mesh sizes. As is evident from Figs. 7, 8 and 9, the influence of the mesh sizes is smaller when the edges are simply supported rather than clamped, and/or when θ° is 0° rather than 45°. When the deflection curve is more complicated, the finer grid is, in general, necessary in order to obtain proper accuracy.

2. Comparison of Experimental and Computed Results

In Figs. 10～13, the examples of experimental results are compared with the results computed with the elastic constants of the plates used in the experiment (Table 4). The satisfactory agreements are recognized in these figures, but in Figs. 12 and 13 there is a tendency that the differences of experimental and computed values of 0°, 15°, 75° and 90° are larger than those of 30°, 45° and 60°. It seems that the difference happened due to the following causes : influence of the overhang of the corners of the plates, disorder of the grain and the uneven distribution of the elastic constants in the plates.

3. Deflection and Moment Distributions

Deflection Distribution

Figs. 14 and 15 show the deflection contour maps of the orthotropic plates with the



Fig. 10. Comparison of computed and observed values of deflection along the center vertical line of the orthotropic square plates with the elastic constants of Table 3 under hydrostatic pressure (top edge free and others simply supported).



Fig. 11. Comparison of computed and observed values (same as Fig. 10).



Fig. 12. Comparison of computed and observed values of deflection along the center vertical line of the orthotropic square plates with the elastic constants of Table 3 under hydrostatic pressure (all edges simply supported).



Fig. 13. Comparison of computed and observed values (same as Fig. 12).

Table 4. Thickness and modulus of elasticity of the plates.

Specimen	t (mm)	E_{Xb} ($\times 10^3$ kg/cm^2)	E_{Yb} ($\times 10^3$ kg/cm^2)	G_{XYb}** ($\times 10^3$ kg/cm^2)
A (0° and 90°)	9.44	94.6	38.4	5.0
B (15° and 75°)	9.45	92.6	37.7	4.8
C (30° and 60°)	9.42	98.1	43.2	4.8
D (45°)	9.40	95.0	41.7	5.2

* As Poisson's ratio the datum of the same species[6] is used. μ_{XYb}: 0.561.
** calculated from $E_{45°b}$ observed.



Fig. 14. Deflection contour map of the orthotropic square plate with the elastic constants of Table 3 and the maximum elastic principal axis parallel to the horizontal edges under hydrostatic pressure (top edge free and others simply supported).



Fig. 15. Deflection contour map of the orthotropic square plate with the elastic constants of Table 3 and the maximum elastic principal axis oriented at 30° to the horizontal edges under hydrostatic pressure (top edge free and others simply supported).



Fig. 16. Deflection contour map of the orthotropic square plate with the elastic constants of Table 3 and the maximum elastic principal axis parallel to the horizontal edges under hydrostatic pressure (all edges simply supported).



Fig. 17. Deflection contour map of the orthotropic square plate with the elastic constants of Table 3 and the maximum elastic axis oriented at 45° to the horizontal edges under hydrostatic pressure (all edges simply supported).

face grain at 0° and 30° to the edges, one edge free, the other edges simply supported, under hydrostatic pressure, and Figs. 16 and 17 show those of the orthotropic plates with the face grain at 0° and 45°, four edges simply supported. From these figures it is evident that: when the principal axes coincide with those of coordinates i. e. face grain 0° and 90°, the patterns of the contour maps are symmetrical with respect to the vertical center line. But when the face grain is not parallel to the edges the maps are not symmetrical. This is one of the typical properties of the orthotropic plates.

From Figs. 12, 13, 16 and 17, it is obvious that, in case of four edges simply supported, the deflection is smaller when the directions of the principal axes do not coincide with the edges than when they coincide.

The method mentioned above can be applied to the practical problems. As an example, the deflection of a plywood panel for the concrete formwork under typical concrete pressure[7] was calculated. Figs. 18 and 19 show the deflection contour maps of the right half of the panel with two vertical studs at one-third points of the long edge of the panel. These figures show that the deflection in the center portion, namely, that in the portion between the two studs, is only about one-third of that in the remaining right and left portions. And in comparison of these two figures, it is recognized that in the plywood panel for concrete formwork the deflection of the panel of horizontal face grain is much smaller than that of vertical one.



Fig. 18. Deflection contour map of 12mm plywood panel for concrete formwork with the face grain parallel to the horizontal edges under concrete pressure[7].
($E_{Xb}=77.3\times10^3$kg/cm², $E_{Yb}=26.3\times10^3$kg/cm², $G_{XYb}=3.8\times10^3$kg/cm²)[8]



Fig. 19. Deflection contour map of 12mm plywood panel for concrete formwork with the face grain parallel to the vertical edges under concrete pressure[7] (see Fig. 18).

Moment Distribution

The distribution of the bending moments can be computed from the deflection distribution by using the equations (5-1)~(5-3) or Fig. 3.

The maximum and minimum principal moments and their directions are derived



Fig. 20. Distribution of moments of the orthotropic square plate with the elastic constants of Table 3 and the maximum elastic principal axis parallel to the horizontal edges under hydrostatic pressure (top edge free and others simply supported).



Fig. 21. Distribution of moments of the orthotropic square plate with the elastic constants of Table 3 and the maximum principal axis parallel to the horizontal edges under hydrostatic pressure (top edge free and others clamped).

from the following formulae:

$$\left.\begin{matrix} M_{max} \\ M_{min} \end{matrix}\right\} = \frac{1}{2}(M_x + M_y) \pm \frac{1}{2}\sqrt{(M_x - M_y)^2 + 4M_{xy}^2}, \quad (17\text{—}1)$$

$$\left.\begin{matrix} \theta_{max} \\ \theta_{min} \end{matrix}\right\} = \frac{1}{2}\tan^{-1}\frac{2M_{xy}}{M_x - M_y} + \frac{m\pi}{2}$$

$$(m: 0 \text{ or integer}). \quad (17\text{—}2)$$

The computed results for the square orthotropic plates of face grain 0°, one edge free, the other edges simply supported, and one edge free, the other edges clamped are shown in Figs. 20 and 21, respectively. The maximum moment of the former is observed near the center but that of the latter on the vertical edges, and moreover, the moment near the edges and that near the center has the opposite sign. The distribution of the principal bending moments as well as their directions of the square plywood plates with the face grain at 30° to the horizontal edges, are shown in Fig. 22. In the figure, the length and the direction of the arrows indicate the quantity and the direction of the maximum and minimum principal moments, respectively. It is one of the most distinctive properties of the orthotropic plates that the moments are somewhat forced to orient to the directions of the axes of the elastic symmetry.



Fig. 22. Distribution of principal moments and their directions of the orthotropic square plate with the elastic constants of Table 3 and the maximum principal axis oriented at 30° to the horizontal edges under hydrostatic pressure (top edge free and others simply supported).

Acknowledgement

The authors wish to thank Mrs. Katsuyama and Miss Kimura for their helpful arrangement of the computation results. The present numerical work was carried out by the computers KDC-II (HITAC 5020) of Kyoto Univ., HITAC 5020 E of Tokyo Univ. and FACOM 230-60 of Kyoto Univ.

Summary

The fundamental equation (6-2) for the deflection of layered orthotropic plates is very difficult to solve rigorously when the axes of elastic symmetry do not coincide with the axes of coordinates ($\beta_{16} \neq 0$, $\beta_{26} \neq 0$). The authors attempted to solve the equation approximately by means of the finite difference method. The computer programs were designed to be able to vary the mesh sizes by merely changing the data cards,

so that the influence of the mesh size of each boundary condition on the accuracy of the approximation could be discussed easily. The experimental results for the deflection of rectangular plywood plates with various grain orientation under hydrostatic pressure, coincide well with the computed results. It seems that most of the problems on the deflection and the moment of the rectangular orthotropic plates can be solved easily by the above method.

摘　　要

直交異方性板に関する基礎微分方程式 (6-2) は，$\beta_{16}=\beta_{26}=0$ すなわち弾性主軸が座標軸に一致している場合で境界条件の簡単な場合には，フーリエ級数を用いて解くことが可能であるが，その他の場合には厳密に解くことが非常に困難もしくは不可能であると考えられる。そこで著者らは，差分法の導入と電算機の利用により，これを近似的に解くことを試みた。プログラミングに際しては，格子粗さの変換をデータカードの交換のみによって容易に行なえるように工夫し，差分格子粗さによる近似精度への影響の検討を容易にした。これにより，境界条件の違いによる正解への収束性の相違を明らかにするとともに，この数値解析法を用いて十分に精度のよい解の得られることを確めた。また，これらの解と実験値との比較を行ない，両者がよく一致することも確認した。この解析方法の確立により，直交異方性長方形板のたわみおよび曲げモーメント分布に関するほとんどすべての問題を容易に解くことができるようになった。

ここでは，多層直交異方性板の数値解析方法およびその検討について例をあげて説明し，かつ2～3の解析結果例を紹介した。

Literature

1) KANTOROVICH, L. V. and V. I. KRYLOV, Approximate Methods of Higher Analysis, 162, Interscience Publisher (1964).
2) AKATSUKA, T., Numerical Computation (in Japanese) (Suuchi-Keisan) 382, Corona Publishing Co., Tokyo, (1967).
3) MAKI, A. C., U. S. Forest Service Research Paper, FPL-87 (1968).
4) MARCH, H. W., U. S. Depart. Agr., FPL. Mimeo. No. 1312 (1942).
5) SAWADA, M. and K. UEDA, Research Bulletin of the College Experiment Forest, Hokkaido Univ., **25**, 61 (1967).
6) UEDA, K., Research Bulletins of the College Experiment Forest, Hokkaido Univ., **26**, 143 (1968).
7) PEURIFOY, R. L., Formwork for Concrete Structures, 30, McGraw-Hill (1964).
8) KAMEDA, Y., Plywood Formwork (in Japanese) (Gohban-Katawaku-Kooho), 24 (1968).

APPENDIX

Flow Chart for Program PIFS (1)



FORTRAN PROGRAM

```
C       PIFS(1)
C       ANALYSIS OF DEFLECTION OF ORTHOTROPIC PLATES
C       WITH A FREE EDGE
C       UNDER HYDROSTATIC PRESSURE
        DIMENSION X(163,208*),Y(163*),YY(163*)
        READ(5,100)LL,LACCUR,MOMOUT
  100   FORMAT(3I2)
        READ(5,101)W,SXL,SYL,JDX,JDY
  101   FORMAT(3F10.0,2I3)
        READ(5,102)EXV,EYV,UXYV
  102   FORMAT(3F10.0)
        DX=JDX
        DY=JDY
        SX=SXL/DX
        SY=SYL/DY
        RS=SY/SX
        M=JDX-1
        N=JDY
        MN=M*N
        MB=M+4
        NB=N+4
        N2=N+2
        N3=N+3
        MNB=MB*NB
        MNU=2*N2+M*N3
        UYXV=UXYV*EYV/EXV
        DO 10 L=1,LL
        READ(5,103)T,EX,EY,G,DEG
  103   FORMAT(5F10.0)
        UYX=UYXV*EXV/EX
        UXY=UYX*EX/EY
        LCHECK=LACCUR
        LXMATR=1
        LPHASE=0
        V=1.0-UXYV*UYXV
        C1=EX/V
        C2=EY/V
        C3=G
        C4=UYX*EX/V
        P=COS(DEG*3.141593/180.0)
        Q=SIN(DEG*3.141593/180.0)
        P4=P**4
        Q4=Q**4
        P2=P**2
        Q2=Q**2
        B11=C1*P4+C2*Q4+2.0*(C4+2.0*C3)*P2*Q2
        B12=(C1+C2-4.0*C3)*P2*Q2+C4*(P4+Q4)
        B16=(C1*P2-C2*Q2)*P*Q-(C4+2.0*C3)*(P2-Q2)*P*Q
        B22=C1*Q4+C2*P4+2.0*(C4+2.0*C3)*P2*Q2
        B26=(C1*Q2-C2*P2)*P*Q+(C4+2.0*C3)*(P2-Q2)*P*Q
        B66=(C1+C2-2.0*C4)*P2*Q2+C3*(P2-Q2)**2
        A=(2.0*B12+4.0*B66)/(B11*RS**2)
        B=B22/(B11*RS**4)
        C=12.0*SX**4/(T**3*B11)
        D=B16/(B11*RS)
        E=B26/(B11*RS**3)
```

```
* DIMENSION
   X(MNU*MNB)
   Y(MNU)
   YY(MNU)
```

```
      F=6.0+4.0*A+6.0*B
      GZ=A+2.0*D+2.0*E
      H=A-2.0*D-2.0*E
      O=-2.0*A-4.0*B
      R=-4.0-2.0*A
      R11=4.0*B11+4.0*B12/RS**2
      R12=-2.0*B12/RS**2
      R13=-2.0*B11
      R14=B16/RS
      R21=4.0*B12+4.0*B22/RS**2
      R22=-2.0*B22/RS**2
      R23=-2.0*B12
      R24=B26/RS
    1 CONTINUE
      DO 12 I=1,MNU
      DO 13 J=1,MNB
      X(I,J)=0.0
   13 CONTINUE
   12 CONTINUE
C     MATRIX OF EQUILIBRIUM EQUATIONS
      DO 20 I=1,N
      DO 21 J=1,M
      NEE=(J-1)*N+I
      K12=NB*(J-1)+I+1
      K13=NB*(J-1)+I+2
      K14=NB*(J-1)+I+3
      K21=NB*J+I
      K22=NB*J+I+1
      K23=NB*J+I+2
      K24=NB*J+I+3
      K25=NB*J+I+4
      K31=NB*(J+1)+I
      K32=NB*(J+1)+I+1
      K33=NB*(J+1)+I+2
      K34=NB*(J+1)+I+3
      K35=NB*(J+1)+I+4
      K41=NB*(J+2)+I
      K42=NB*(J+2)+I+1
      K43=NB*(J+2)+I+2
      K44=NB*(J+2)+I+3
      K45=NB*(J+2)+I+4
      K52=NB*(J+3)+I+1
      K53=NB*(J+3)+I+2
      K54=NB*(J+3)+I+3
      X(NEE,K12)=-D
      X(NEE,K13)=1.0
      X(NEE,K14)=D
      X(NEE,K21)=-E
      X(NEE,K22)=GZ
      X(NEE,K23)=R
      X(NEE,K24)=H
      X(NEE,K25)=E
      X(NEE,K31)=B
      X(NEE,K32)=O
      X(NEE,K33)=F
      X(NEE,K34)=O
      X(NEE,K35)=B
      X(NEE,K41)=E
      X(NEE,K42)=H
      X(NEE,K43)=R
```

FORTRAN representation	Specification
EX	E_{xb}
EXV	E_{xv}
EY	E_{yb}
EYV	E_{yv}
G	G_{xyb}
JDX	a/h_x
JDY	b/h_y
LACCUR	see Flow Chart
LL	number of data cards for the plates
MOMOUT	see Flow Chart
SXL	a
SYL	b
T	t
UXY	μ_{XYb}
UXYV	μ_{XYv}
W	specific gravity of the water
X (array)	matrix of the left-hand side of the simultaneous linear equations
Y (array)	matrix of the left-hand side of the simultaneous linear equations

```
      X(NEE,K44)=GZ
      X(NEE,K45)=-E
      X(NEE,K52)=D
      X(NEE,K53)=1.0
      X(NEE,K54)=-D
   21 CONTINUE
   20 CONTINUE
C     LEFT AND RIGHT-SIDE EDGES     MX=0
      DO 24 II=1,N+1
      I=II-1
      NEM=MN+II
      MX11=I+1
      MX12=I+2
      MX13=I+3
      MX21=NB+I+1
      MX22=NB+I+2
      MX23=NB+I+3
      MX31=2*NB+I+1
      MX32=2*NB+I+2
      MX33=2*NB+I+3
      X(NEM,MX11)=R14
      X(NEM,MX12)=R13
      X(NEM,MX13)=-R14
      X(NEM,MX21)=R12
      X(NEM,MX22)=R11
      X(NEM,MX23)=R12
      X(NEM,MX31)=-R14
      X(NEM,MX32)=R13
      X(NEM,MX33)=R14
   24 CONTINUE
      DO 25 JJ=1,N+1
      J=JJ-1
      NEM=MN+N+1+JJ
      MX11=NB*(M+1)+J+1
      MX12=NB*(M+1)+J+2
      MX13=NB*(M+1)+J+3
      MX21=NB*(M+2)+J+1
      MX22=NB*(M+2)+J+2
      MX23=NB*(M+2)+J+3
      MX31=NB*(M+3)+J+1
      MX32=NB*(M+3)+J+2
      MX33=NB*(M+3)+J+3
      X(NEM,MX11)=R14
      X(NEM,MX12)=R13
      X(NEM,MX13)=-R14
      X(NEM,MX21)=R12
      X(NEM,MX22)=R11
      X(NEM,MX23)=R12
      X(NEM,MX31)=-R14
      X(NEM,MX32)=R13
      X(NEM,MX33)=R14
   25 CONTINUE
C     UPPER AND LOWER-SIDE EDGES     MY=0
      DO 28 I=1,M
      NEM=MN+2*(N+1)+I
      MY11=NB*I+2
      MY12=NB*I+3
      MY13=NB*I+4
      MY21=NB*(I+1)+2
      MY22=NB*(I+1)+3
```

```
      MY23=NB*(I+1)+4
      MY31=NB*(I+2)+2
      MY32=NB*(I+2)+3
      MY33=NB*(I+2)+4
      X(NEM,MY11)=R24
      X(NEM,MY12)=R23
      X(NEM,MY13)=-R24
      X(NEM,MY21)=R22
      X(NEM,MY22)=R21
      X(NEM,MY23)=R22
      X(NEM,MY31)=-R24
      X(NEM,MY32)=R23
      X(NEM,MY33)=R24
   28 CONTINUE
      DO 29 J=1,M
      NEM=MN+2*(N+1)+M+J
      MY11=NB*(J+1)-2
      MY12=NB*(J+1)-1
      MY13=NB*(J+1)
      MY21=NB*(J+2)-2
      MY22=NB*(J+2)-1
      MY23=NB*(J+2)
      MY31=NB*(J+3)-2
      MY32=NB*(J+3)-1
      MY33=NB*(J+3)
      X(NEM,MY11)=R24
      X(NEM,MY12)=R23
      X(NEM,MY13)=-R24
      X(NEM,MY21)=R22
      X(NEM,MY22)=R21
      X(NEM,MY23)=R22
      X(NEM,MY31)=-R24
      X(NEM,MY32)=R23
      X(NEM,MY33)=R24
   29 CONTINUE
C     FREE EDGE   VY=0
      Q21=B22/RS**3
      Q22=2.0*B16
      Q23=(B12+4.0*B66)/RS+4.0*B26/RS**2
      Q24=-(B12+4.0*B66)/RS+4.0*B26/RS**2
      Q25=4.0*B16+8.0*B26/RS**2
      Q26=2.0*(B12+4.0*B66)/RS+2.0*B22/RS**3
      DO 70 I=1,M
      NEV=MN+2*(N+1)+2*M+I
      MV13=NB*(I-1)+3
      MV22=NB*I+2
      MV23=NB*I+3
      MV24=NB*I+4
      MV31=NB*(I+1)+1
      MV32=NB*(I+1)+2
      MV34=NB*(I+1)+4
      MV35=NB*(I+1)+5
      MV42=NB*(I+2)+2
      MV43=NB*(I+2)+3
      MV44=NB*(I+2)+4
      MV53=NB*(I+3)+3
      X(NEV,MV13)=-Q22
      X(NEV,MV22)=-Q24
      X(NEV,MV23)=Q25
      X(NEV,MV24)=-Q23
```

```
      X(NEV,MV31)=Q21
      X(NEV,MV32)=-Q26
      X(NEV,MV34)=Q26
      X(NEV,MV35)=-Q21
      X(NEV,MV42)=Q23
      X(NEV,MV43)=-Q25
      X(NEV,MV44)=Q24
      X(NEV,MV53)=Q22
   70 CONTINUE
C     LEFT UPPER CORNER
      NCL=MNU-1
      X(NCL,1)=1.0
      X(NCL,2*NB+1)=1.0
C     RIGHT UPPER CORNER
      NNI=MNB-3*NB
      NNC=MNB-NB
      X(MNU,NNI+1)=1.0
      X(MNU,NNC+1)=1.0
      IF(LXMATR.GE.2) GO TO 3
C     MATRIX OF Y  (RIGHT HAND SIDE), HYDROSTATIC PRESSURE
      DO 35 I=1,MNU
      Y(I)=0.0
   35 CONTINUE
      CW=C*W*SY
      DO 36 J=1,M
      DO 37 I=1,N
      II=(J-1)*N+I
      DE=I-1
      Y(II)=DE*CW
   37 CONTINUE
   36 CONTINUE
C     MATRIX ARRANGEMENT OF Y AND X
      DO 38 I=1,MNU
      YY(I)=Y(I)
   38 CONTINUE
      CALL XMATRX(X,M,N,NB,MNU,MNB)
C     COMPUTATION
      CALL LINSW(X,YY,MNU,0)
C     RESULT OUTPUT
    4 WRITE(6,200)
  200 FORMAT(1H1,10X,44HANALYSIS OF DEFLECTION OF ORTHOTROPIC PLATES/1H0
     1,10X,51HWITH ONE FREE EDGE AND THREE SIMPLY SUPPORTED EDGES)
      WRITE(6,201)
  201 FORMAT(1H0,10X,49HTHE GRAIN OF THE FACE PLIES INCLINED TO THE EDGE
     1S/1H0,10X,26HUNDER HYDROSTATIC PRESSURE//)
      WRITE(6,202)T,EX,EY,G,DEG,UXY,UYX,V
  202 FORMAT(1H0,5X,2HT=E12.5,8X,3HEX=E12.5,7X,3HEY=E12.5,7X,2HG=E12.5/1
     1H0,5X,4HDEG=E12.5,6X,4HUXY=E12.5,6X,4HUYX=E12.5,6X,2HV=E12.5/)
      WRITE(6,203)W,SXL,SYL,SX,SY,EXV,EYV,UXYV,UYXV
  203 FORMAT(1H0,5X,2HW=E12.5,8X,4HSXL=E12.5,6X,4HSYL=E12.5,6X,3HSX=E12.
     15,7X,3HSY=E12.5/1H0,5X,4HEXV=E12.5,6X,4HEYV=E12.5,6X,5HUXYV=E12.5,
     25X,5HUYXV=E12.5/)
      REXY=EX/EY
      WRITE(6,204)REXY,P,Q,B11,B12,B16,B22,B26,B66
  204 FORMAT(1H0,5X,6HEX/EY=E12.5,4X,4HCOS=E12.5,6X,4HSIN=E12.5/1H0,5X,4
     1HB11=E12.5,6X,4HB12=E12.5,6X,4HB16=E12.5/1H0,5X,4HB22=E12.5,6X,4HB
     226=E12.5,6X,4HB66=E12.5///)
      LPHASE=LPHASE+1
      WRITE(6,211)LPHASE
  211 FORMAT(1H ,5HPHASEI2)
```

```
      DIMENSION JMAT(18),YOM(18,18),JFM(18)
      DO 40 I=1,N+2
      DO 41 J=1,M
      II=N+2+(N+3)*(J-1)+I
      YOM(I,J)=YY(II)
   41 CONTINUE
   40 CONTINUE
      DO 42 J=1,18
      JMAT(J)=J
      JFM(J)=J-2
   42 CONTINUE
      WRITE(6,300)(JMAT(J),J=1,M)
  300 FORMAT(1H0,9I12)
      WRITE(6,350)((YOM(I,J),J=1,M),I=1,2)
  350 FORMAT(1H0,2X,9F12.5)                                            * M
      WRITE(6,301)(JFM(I),(YOM(I,J),J=1,M),I=3,N+2)
  301 FORMAT(1H0,I2,9F12.5)                                            * M
C     OUTER POINT RESULT OUTPUT
      WRITE(6,302)(YY(I),I=1,N+2)
  302 FORMAT(//1H0,5X,17HLEFT OUTER POINTS/1H0,2X,12F10.5//)
      MER=MNU-N-1
      WRITE(6,303)(YY(I),I=MER,MNU)
  303 FORMAT(1H0,5X,18HRIGHT OUTER POINTS/1H0,2X,12F10.5//)
      MEB=N+2+(N+3)*M
      MEL=2*N+5
      WRITE(6,305)(YY(I),I=MEL,MEB,N+3)
  305 FORMAT(1H0,5X,19HBOTTOM OUTER POINTS/1H0,2X,9F12.5//)
      IF(LXMATR.GE.3) GO TO 1
      LXMATR=2
      IF(LCHECK)15,15,1
C     RESULT CHECK
      DIMENSION JR(163),YZ(163)
    3 CONTINUE
      CALL XMATRX(X,M,N,NB,MNU,MNB)
      DO 65 I=1,MNU
      JR(I)=I
      ZZ=0.0
      DO 66 J=1,MNU
      ZZ=ZZ+YY(J)*X(I,J)
   66 CONTINUE
      YZ(I)=Y(I)-ZZ
      WRITE(6,620)JR(I),ZZ,Y(I),YZ(I)
  620 FORMAT(1H0,5X,I3,3E20.5)
   65 CONTINUE
      IF(LCHECK.EQ.1)GO TO 15
      LCHECK=LCHECK-1
      CALL LINSW(X,YZ,MNU,0)
      DO 67 I=1,MNU
      YY(I)=YY(I)+YZ(I)
   67 CONTINUE
      LXMATR=3
      GO TO 4
   15 CONTINUE
      IF(MOMOUT.EQ.0) GO TO 5
C     MOMENT MX,MY,MXY,MAX OUTPUT
      DIMENSION YM(16,16),OMX(12,12),OMY(12,12),OMXY(12,12)
      DO 80 I=1,N+4
      DO 81 J=1,M+4
      YM(I,J)=0.0
   81 CONTINUE
```

```
   80 CONTINUE
C     DEFLECTION COPY OF INNER POINTS
      DO 82 I=1,N+2
      DO 83 J=1,M
      JJ=J+2
      IY=N+2+(N+3)*(J-1)+I
      YM(I,JJ)=YY(IY)
   83 CONTINUE
   82 CONTINUE
C     DEFLECTION COPY OF OUTER POINTS
      DO 84 I=1,N+2
      YM(I,1)=YY(I)
      IY=MNU-N-2+I
      YM(I,M+4)=YY(IY)
   84 CONTINUE
      DO 85 J=1,M
      IY=N+2+(N+3)*J
      YM(N+4,J+2)=YY(IY)
   85 CONTINUE
C     BOTTOM CORNERS   LEFT AND RIGHT.....MX=0
      YM(N+4,1)=YM(N+2,1)-YM(N+2,3)+YM(N+4,3)
      YM(N+4,M+4)=-YM(N+2,M+2)+YM(N+4,M+2)+YM(N+2,M+4)
      R61=4.0*B16+4.0*B26/RS**2
      R62=-2.0*B26/RS**2
      R63=-2.0*B16
      R64=B66/RS
      DO 86 II=1,N+1
      DO 87 JJ=1,M+2
      I=II+2
      J=JJ+1
      OMX(II,JJ)=T**3/(24.0*SX**2)*(R14*YM(I-1,J-1)+R13*YM(I,J-1)-R14*YM
     1(I+1,J-1)+R12*YM(I-1,J)+R11*YM(I,J)+R12*YM(I+1,J)-R14*YM(I-1,J+1)+
     2R13*YM(I,J+1)+R14*YM(I+1,J+1))
      OMY(II,JJ)=T**3/(24.0*SX**2)*(R24*YM(I-1,J-1)+R23*YM(I,J-1)-R24*YM
     1(I+1,J-1)+R22*YM(I-1,J)+R21*YM(I,J)+R22*YM(I+1,J)-R24*YM(I-1,J+1)+
     2R23*YM(I,J+1)+R24*YM(I+1,J+1))
      OMXY(II,JJ)=T**3/(24.0*SX**2)*(R64*YM(I-1,J-1)+R63*YM(I,J-1)-R64*Y
     1M(I+1,J-1)+R62*YM(I-1,J)+R61*YM(I,J)+R62*YM(I+1,J)-R64*YM(I-1,J+1)
     2+R63*YM(I,J+1)+R64*YM(I+1,J+1))
   87 CONTINUE
   86 CONTINUE
      WRITE(6,600)
  600 FORMAT(1H1,10X,25HMOMENT CALCULATION.....MX/)
      WRITE(6,601)(JMAT(J),J=1,M)
  601 FORMAT(1H0,11X,9I10)
      WRITE(6,602)(JMAT(I),(OMX(I,J),J=1,M+2),I=1,N+1)
  602 FORMAT(1H0,I2,11F10.5)                                  * M+2
      WRITE(6,603)
  603 FORMAT(1H1,10X,25HMOMENT CALCULATION.....MY/)
      WRITE(6,601)(JMAT(J),J=1,M)
      WRITE(6,602)(JMAT(I),(OMY(I,J),J=1,M+2),I=1,N+1)
      WRITE(6,604)
  604 FORMAT(1H1,10X,26HMOMENT CALCULATION.....MXY/)
      WRITE(6,601)(JMAT(J),J=1,M)
      WRITE(6,602)(JMAT(I),(OMXY(I,J),J=1,M+2),I=1,N+1)
C     MAX.AND MIN. MOMENT
      DIMENSION OMA(13,13),OMI(13,13),OMB(13,13),BT(13,13)
      DO 90 I=1,N+1
      DO 91 J=1,M+2
      BT(I,J)=0.5*ATAN(2.0*OMXY(I,J)/(OMX(I,J)-OMY(I,J)))*180.0/3.141593
```

```
      BE=BT(I,J)*3.141593/180.0
      OMB(I,J)=OMY(I,J)*SIN(BE)**2+OMX(I,J)*COS(BE)**2+OMXY(I,J)*SIN(2.0
     1*BE)
      ROT=(OMX(I,J)-OMY(I,J))**2+4.0*OMXY(I,J)**2
      OMA(I,J)=0.5*(OMX(I,J)+OMY(I,J))+0.5*SQRT(ROT)
      OMI(I,J)=0.5*(OMX(I,J)+OMY(I,J))-0.5*SQRT(ROT)
   91 CONTINUE
   90 CONTINUE
      WRITE(6,610)
  610 FORMAT(1H1,10X,44HDIRECTION OF MAX OR MIN MOMENT.....BT DEGREE/)
      WRITE(6,601)(JMAT(J),J=1,M)
      WRITE(6,602)(JMAT(I),(BT(I,J),J=1,M+2),I=1,N+1)
      WRITE(6,611)
  611 FORMAT(1H1,10X,27HMAX. OR MIN. MOMENT.....MBT/)
      WRITE(6,601)(JMAT(J),J=1,M)
      WRITE(6,602)(JMAT(I),(OMB(I,J),J=1,M+2),I=1,N+1)
      WRITE(6,612)
  612 FORMAT(1H1,10X,22HMAXIMUM MOMENT.....MAX/)
      WRITE(6,601)(JMAT(J),J=1,M)
      WRITE(6,602)(JMAT(I),(OMA(I,J),J=1,M+2),I=1,N+1)
      WRITE(6,613)
  613 FORMAT(1H1,10X,22HMINIMUM MOMENT.....MIN/)
      WRITE(6,601)(JMAT(J),J=1,M)
      WRITE(6,602)(JMAT(I),(OMI(I,J),J=1,M+2),I=1,N+1)
    5 CONTINUE
   10 CONTINUE
      STOP
      END

      SUBROUTINE XMATRX(X,M,N,NB,MNU,MNB)
      DIMENSION X(MNU,MNB)
      DO 60 I=1,MNU
C     LEFT OUTER POINTS ..... SAME
C     INNER POINTS
      DO 63 JJ=1,M
      DO 64 II=1,N+2
      NU5=N+2+(N+3)*(JJ-1)+II
      NB5=NB*(JJ+1)+II
      X(I,NU5)=X(I,NB5)
   64 CONTINUE
      NU4=N+JJ*(N+3)+2
      NB4=NB*(JJ+2)
      X(I,NU4)=X(I,NB4)
   63 CONTINUE
C     RIGHT OUTER POINTS
      DO 65 J=1,N+2
      NU2=MNU-N+J-2
      NB2=MNB-NB+J
      X(I,NU2)=X(I,NB2)
   65 CONTINUE
   60 CONTINUE
      RETURN
      END

      SUBROUTINE LINSW(A,AN,N1,NSTOP)
C     SWEEP OUT METHOD.....SOLUTION FOR LINEAR EQUATIONS
C     *** OMITTED ***
```